

## Ссылки R1C1

Это глава из книги Билла Джелена [Гуру Excel расширяют горизонты: делайте невозможное с Microsoft Excel](#).

Задача: ссылки в стиле R1C1 полезны в нескольких ситуациях, в частности, при использовании VBA, функции ДВССЫЛ (см., например, последний раздел заметки [Ссылка на другой лист с помощью ДВССЫЛ](#)) и условного форматирования. Поэтому краткое знакомство с ними будет полезно.

До появления платформы IBM PC существовало несколько популярных приложений электронных таблиц. Это были и VisiCalc и Quattro Pro и Multiplan. Впервые стиль A1 именования ячеек был представлен в VisiCalc. Но этот продукт быстро проиграл конкурентную гонку. Multiplan выпускался компанией Microsoft до выхода Excel. В этом продукте для адресации ячеек использовался формат R1C1, который с тех пор доступен и в Excel. Но именно Лотус 1-2-3 вырвался на вершину славы сразу же после выхода в 1982 г. и стал доминировать на рынке приложений электронных таблиц для ПК. Не в последнюю очередь благодаря удобной системе именования ячеек A1.<sup>1</sup>

Во время войн электронных таблиц, Microsoft осознала, что большинство пользователей голосуют за стиль ссылок A1, и чтобы конкурировать, ей придется делать вид, что в Excel используются ссылки A1. Конечно, Excel только делает вид, что использует A1. В действительности «за кулисами» работают R1C1. Если вы не верите мне, пройдите по меню *ФАЙЛ* → *Параметры*, перейдите на вкладку *Формулы*, и в области *Работа с формулами* поставьте галочку *Стиль ссылок R1C1* (рис. 1).




Рис. 1. Вы всего в одном клике от стиля ссылок R1C1

Обратите внимание, что я не призываю вас перейти на стиль ссылок R1C1. Я не такой сумасшедший, как Microsoft, которая при переходе с Excel 2003 на Excel 2007 полностью заменила привычное меню. Вместо этого, я предлагаю вам изучить стиль R1C1, чтобы применять его, когда это уместно.

Решение: стиль ссылок R1C1 содержит буквы R (row – строка) и C (column – столбец). Без каких-либо чисел (т.е. в виде RC) R означает «та же строка, в которой введена формула» и C означает «тот же столбец, в который введена формула». Таким образом, простейшая ссылка в стиле R1C1 – это =RC. Если вы находитесь в ячейке C10 и введете =RC, вы имеете в виду ячейку C10. Осторожно! Это приведет к возникновению ошибки – циклической ссылки.

Когда в дополнение к буквам R и C используются числа в квадратных скобках, вы ссылаетесь на ячейку, отстоящую от данной на указанное число ячеек. Например, формула =RC[-1] введенная в C10 ссылается на B10, а =R[10]C в C10 – на C20. Вы можете одновременно изменить, и строку, и столбец. Например, формула =R[1]C[1] введенная в C10 ссылается на D11.

Примечание: для строк положительные числа направляют вниз листа, отрицательные – вверх. Для столбцов положительные числа направляют вправо, отрицательные – влево.

<sup>1</sup> Цитируется по изданию Пол Киммел и др. Excel 2003 и VBA. Справочник программиста.

Если вы хотите построить формулу для расчета рентабельности (GP) в колонке I, вы должны разделить значение в соответствующей строке столбца H на значение в столбце F (рис. 2). Видно, что формулы в столбце I разные в каждой строке.

	F	G	H	I
1	Revenue	COGS	Profit	GP%
2	22810	10220	12590	=H2/F2
3	2257	984	1273	=H3/F3
4	18552	7872	10680	=H4/F4
5	9152	4088	5064	=H5/F5
6	21730	9840	11890	=H6/F6
7	8456	3388	5068	=H7/F7
8	16416	6776	9640	=H8/F8
9	21438	9198	12240	=H9/F9
10	6267	2541	3726	=H10/F10

	6	7	8	9
1	Revenue	COGS	Profit	GP%
2	22810	10220	12590	=RC[-1]/RC[-3]
3	2257	984	1273	=RC[-1]/RC[-3]
4	18552	7872	10680	=RC[-1]/RC[-3]
5	9152	4088	5064	=RC[-1]/RC[-3]
6	21730	9840	11890	=RC[-1]/RC[-3]
7	8456	3388	5068	=RC[-1]/RC[-3]
8	16416	6776	9640	=RC[-1]/RC[-3]
9	21438	9198	12240	=RC[-1]/RC[-3]
10	6267	2541	3726	=RC[-1]/RC[-3]

Рис. 2. Со ссылками в стиле A1 (верхний рисунок) формулы в каждой ячейке разные, со ссылками в стиле R1C1 (нижний рисунок) формулы одинаковые.

А что если перейти к ссылкам в стиле R1C1? Вам нужно разделить значение в столбце на один левее текущего, на значение в столбце на три левее текущего. Вы можете использовать одну и ту же формулу, независимо от того, в какой строке вы находитесь: =RC[-1]/RC[-3].

А что с абсолютными ссылками в стиле R1C1? Формулы в столбце J (рис. 3) гарантируют, что вы всегда используете налоговую ставку из ячейки M1. Для указания абсолютной ссылки в стиле R1C1, опустите квадратные скобки и укажите не смещение от текущей строки, а абсолютные номера строки и столбца. Например, ячейка \$M\$1 – это 13-й столбец строки 1, поэтому, используя R1C13.

	F	G	H	I	J	K	L	M
1	Revenue	COGS	Profit	GP%	Sales Tax		Tax Rate	0,07
2	22810	10220	12590	=H2/F2	=F2*\$M\$1			
3	2257	984	1273	=H3/F3	=F3*\$M\$1			
4	18552	7872	10680	=H4/F4	=F4*\$M\$1			
5	9152	4088	5064	=H5/F5	=F5*\$M\$1			
6	21730	9840	11890	=H6/F6	=F6*\$M\$1			
7	8456	3388	5068	=H7/F7	=F7*\$M\$1			
8	16416	6776	9640	=H8/F8	=F8*\$M\$1			
9	21438	9198	12240	=H9/F9	=F9*\$M\$1			
10	6267	2541	3726	=H10/F10	=F10*\$M\$1			
11	2401	1022	1379	=H11/F11	=F11*\$M\$1			

	6	7	8	9	10	11	12	13
1	Revenue	COGS	Profit	GP%	Sales Tax		Tax Rate	0,07
2	22810	10220	12590	=RC[-1]/RC[-3]	=RC[-4]*R1C13			
3	2257	984	1273	=RC[-1]/RC[-3]	=RC[-4]*R1C13			
4	18552	7872	10680	=RC[-1]/RC[-3]	=RC[-4]*R1C13			
5	9152	4088	5064	=RC[-1]/RC[-3]	=RC[-4]*R1C13			
6	21730	9840	11890	=RC[-1]/RC[-3]	=RC[-4]*R1C13			
7	8456	3388	5068	=RC[-1]/RC[-3]	=RC[-4]*R1C13			
8	16416	6776	9640	=RC[-1]/RC[-3]	=RC[-4]*R1C13			
9	21438	9198	12240	=RC[-1]/RC[-3]	=RC[-4]*R1C13			
10	6267	2541	3726	=RC[-1]/RC[-3]	=RC[-4]*R1C13			
11	2401	1022	1379	=RC[-1]/RC[-3]	=RC[-4]*R1C13			

Рис. 3. Абсолютные ссылки в стиле A1 (верхний рисунок) и R1C1 (нижний рисунок)

Резюме: ссылки R1C1 пригодятся вам при записи макросов, программировании VBA, использовании функции ДССЫЛ и условном форматировании.